Document Number: DSP0207

Date: 2009-07-29

Version: 1.0.0

WBEM URI Mapping Specification

Document Type: Specification

Document Status: DMTF Standard

Document Language: E

CONTENTS

Figures

55

Foreword

The *WBEM URI Mapping Specification* (DSP0207) was prepared by the DMTF WBEM Modeling Working Group.

DMTF is a not-for-profit association of industry members dedicated to promoting enterprise and systems management and interoperability.

The Distributed Management Task Force, Inc. (DMTF), developer of CIM, is the industry organization leading the development, adoption, and interoperability of management specifications and initiatives for enterprise and Internet environments.

This specification was processed and approved by the DMTF. Approval of this specification does not necessarily imply that all members voted for approval.

Acknowledgments

The authors wish to acknowledge the following people.

Contributors:

- Jim Davis – WBEM Solutions, Inc
- Paul von Beheren – Sun Microsystems, Inc.
- David Black – EMC
- Denise Eckstein – Hewlett-Packard Company
- George Ericson – EMC
- Steve Hand – Veritas
- Andreas Maier – IBM

Introduction

This specification defines the WBEM Universal Resource Identifier (URI). The WBEM URI is used in WBEM protocols to identify several kinds of CIM objects.

A URI (as defined by the IETF in RFC 3986) is a compact string representation for a resource available via the Internet. This specification defines the subset of the URI syntax that is specific to WBEM.

The reader is expected to have a working knowledge of URI and WBEM.

WBEM URI Mapping Specification

1 Scope

This document defines the Universal Resource Identifier (URI) format for WBEM protocols. A WBEM URI is a compact string of characters for identifying a CIM element. This document defines a mapping of CIM naming, as defined in the CIM Infrastructure Specification (DSP0004), to the URI syntax, as defined in RFC 3986.

The URI Generic Syntax standard and the Universal Resource Locators standards by the IETF provide a framework for identifying resources.

2 Normative References

The following referenced documents are indispensable for the application of this document. For dated references, only the edition cited applies. For undated references, the latest edition of the referenced document (including any amendments) applies.

2.1 Approved References

DMTF DSP0004, *Common Information Model (CIM) Specification 2.5*,
http://www.dmtf.org/standards/published_documents/DSP0004_2.5.pdf

DMTF DSP0200, *CIM Operations over HTTP 1.3*,
http://www.dmtf.org/standards/published_documents/DSP0200_1.3.pdf

IETF RFC 1034, *Domain Names — Concepts and Facilities*, November 1987,
http://www.ietf.org/rfc/rfc1034.txt

IETF RFC 1123, *Requirements for Internet Hosts — Applications and Support*, October 1989,
http://www.ietf.org/rfc/rfc1123.txt

IETF RFC 2373, *IP Version 6 Addressing Architecture*, July 1998,
http://www.ietf.org/rfc/rfc2373.txt

IETF RFC 2717, *Registration Procedures for URL Scheme Names*, November 1999,
http://www.ietf.org/rfc/rfc2717.txt

IETF RFC 2718, *Guidelines for new URL Schemes*, November 1999,
http://www.ietf.org/rfc/rfc2718.txt

IETF RFC 3986, *Uniform Resource Identifiers (URI): Generic Syntax*, January 2005,
http://www.ietf.org/rfc/rfc3986.txt

IETF RFC 5234, *Augmented BNF for Syntax Specifications: ABNF*, January 2008,
http://www.ietf.org/rfc/rfc5234.txt

2.2 Other References

ISO/IEC Directives, Part 2, *Rules for the structure and drafting of International Standards*,
http://isotc.iso.org/livelink/livelink.exe?func=ll&objId=4230456&objAction=browse&sort=subtype

3 Terms and Definitions

For the purposes of this document, the following terms and definitions apply.

3.1
can
used for statements of possibility and capability, whether material, physical, or causal

3.2
cannot
used for statements of possibility and capability, whether material, physical or causal

3.3
conditional
indicates requirements to be followed strictly in order to conform to the document when the specified conditions are met

3.4
mandatory
indicates requirements to be followed strictly in order to conform to the document and from which no deviation is permitted

3.5
may
indicates a course of action permissible within the limits of the document

3.6
need not
indicates a course of action permissible within the limits of the document

3.7
optional
indicates a course of action permissible within the limits of the document

3.8
shall
indicates requirements to be followed strictly in order to conform to the document and from which no deviation is permitted

3.9
shall not
indicates requirements to be followed strictly in order to conform to the document and from which no deviation is permitted

3.10
should
indicates that among several possibilities, one is recommended as particularly suitable, without mentioning or excluding others, or that a certain course of action is preferred but not necessarily required

3.11
should not
indicates that a certain possibility or course of action is deprecated but not prohibited

4 Symbols and Abbreviated Terms

The following symbols and abbreviations are used in this document.

4.1
ABNF
Augmented Backus-Naur Form

4.2
BNF
Backus-Naur Form

4.3
CIM
Common Information Model

4.4
URI
Universal Resource Identifier

4.5
WBEM
Web-Based Enterprise Management

5 WBEM URI

5.1 General

This section specifies the WBEM URI.

The WBEM URI can be used to reference the following kinds of CIM objects:

- CIM namespaces
- CIM classes
- CIM instances
- CIM qualifier types

The following two formats of the WBEM URI are defined:

- Untyped WBEM URI is a format compatible with the CIM Object Name format defined by DSP0004.
- Typed WBEM URI is a format that includes the data type of the key properties, as well as the kind of CIM object referenced.

The Untyped WBEM URI is included to support legacy use of the ObjectName syntax defined in the DMTF DSP0004. For some applications, the Untyped WBEM URI encoding lacks sufficient type information to allow the URI to be processed unambiguously. For this reason, use of the Typed WBEM URI is preferred over the Untyped WBEM URI.

The WBEM URI consists of the components defined in the following subclauses. Specifications that use the WBEM URI shall define the components that are required in their usage of the WBEM URI. The components should be referenced using the BNF production name defined in this document.

Figure 1 illustrates the Mapping of the CIM Object Name to the IETF URI.



Figure 1 – WBEM URI as a Mapping of CIM Object Name to IETF URI

5.2 Namespace Type or Scheme

The Namespace Type is defined as <scheme> in section 3.1 of RFC 3986.

Each WBEM protocol specification shall define the namespace types it supports.

WBEM defined namespace types shall be in the form of <protocol>.wbem or <protocol>.wbems where protocol is the name of the WBEM protocol. The <protocol>.wbems shall be used for the secure version of the WBEM protocol. For example, for CIM-XML the protocol would be cimxml.wbem and cimxml.wbems, respectively. A WBEM protocol may also support additional namespace types. For example, the namespace types “http” and/or “https” would also be valid namespace types for the CIM-XML protocol.

5.3 Authority

Refer to section 3.2 of RFC 3986 for the definition of the authority.

5.4 Namespace Name

The <namespace> may contain the “/” character but shall not begin or end with a “/”.

5.5 Model Path Encoding

Two model path encodings are defined in this specification:

- untyped_modelpath
- typed_modelpath

If the reference is to a CIM Namespace, the model path encoding shall not be present.

If the reference is to a CIM Class, the model path encoding shall only include the className.

If the reference is to an instance, the model path encoding shall include both the className and a key_value_pair or a typed_key_value_pair for all key properties of the instance.

If the reference is to a Qualifier Type, the model path shall only include the qualifierName.

Examples of Untyped Key value pairs:

- stringProperty ="This is a string value with a double quote \" included in the string."
- char16Property='a'
- char16Property='\x32'
- booleanProperty=TRUE
- integerProperty=1000
- integerProperty=100101B
- datetimeProperty="19980525133015.000000-300"
- datetimeProperty="00000001132312.000000:000"
- CIM_IndicationSubscription.Filter="CIM_IndicationFilter.SystemCreationClassName=\"CIM_ComputerSystem\",SystemName=\"server001.acme.com\",CreationClassName=\"CIM_IndicationHandlerCIMXML\",Name=\"Filter01\"",Handler="CIM_IndicationHandlerCIMXML.SystemCreationClassName=\"CIM_ComputerSystem\",SystemName=\"server001.acme.com\",CreationClassName=\"CIM_IndicationHandlerCIMXML\",Name=\"Handler01\""
- ACME_DoubleQuoteExample.Antecedent="ACME_AntecedentClass.Name =\"Before double quote \\\" after double quote\"",ACME_DependentClass.Name=\"Before backslash \\\\ after backslash\""

Examples of Typed Key value pairs:

- stringProperty = (string)"This is a string value with a double quote \" included in the string."
- char16Property=(char16)'a'
- char16Property=(char16)'\x32'
- booleanProperty=(boolean)TRUE
- integerProperty=(sint32)1000
- integerProperty=(uint32)100101B
- integerProperty=(sint64)-12310
- datetimeProperty=(datetime)"19980525133015.000000-300"
- datetimeProperty=(datetime)"00000001132312.000000:000"

- CIM_IndicationSubscription.Filter=(reference)"CIM_IndicationFilter.SystemCreationClassName =(string)\"CIM_ComputerSystem\",SystemName=(string)\"server001.acme.com\",CreationClass Name=(string)\"CIM_IndicationHandlerCIMXML\",Name=(string)\"Filter01\"",Handler=(reference)"CIM_IndicationHandlerCIMXML.SystemCreationClassName=(string)\"CIM_ComputerSystem\" ,SystemName=(string)\"server001.acme.com\",CreationClassName=(string)\"CIM_IndicationHa ndlerCIMXML\",Name=(string)\"Handler01\""
- ACME_DoubleQuoteExample.Antecedent=(reference)"ACME_AntecedentClass.Name =(string)\"Before double quote \\\" after double quote\"",ACME_DependentClass.Name=(string)\"Before backslash \\\\ after backslash\""

5.6 Collected BNF for WBEM URI

The DMTF WBEM URI BNF is a conformant subset of the BNF defined in RFC 3986. To minimize the impact of future changes, implementations should be designed to reject, as invalid or unsupported, WBEM URIs that do not conform to an approved version of the DMTF WBEM URI BNF.

This BNF conforms to ABNF as specified by RFC 5234.

The defining specification shall define the allowable character encodings (for example, UTF-8 or UCS-2) for a WBEM URI.

```
WBEM-URI              = WBEM-URI-TypedPath /
                        WBEM-URI-UntypedPath

WBEM-URI-TypedPath    = WBEM-URI-TypedNamespacePath /
                        WBEM-URI-TypedClassPath /
                        WBEM-URI-TypedInstancePath /
                        WBEM-URI-TypedQualifierTypePath

WBEM-URI-UntypedPath  = WBEM-URI-UntypedNamespacePath /
                        WBEM-URI-UntypedClassPath /
                        WBEM-URI-UntypedInstancePath

// Note: The production rules for className and qualifierName are
// defined in Appendix A (MOF Syntax Grammar Description) of DMTF
// DSP0004, CIM Infrastructure Specification.

WBEM-URI-TypedNamespacePath      = namespacePath "/(namespace)"
WBEM-URI-TypedClassPath          = namespacePath "/(class)" className
WBEM-URI-TypedInstancePath       = namespacePath "/(instance)"
                                   className "." typed_key_value_pairs
WBEM-URI-TypedQualifierTypePath  = namespacePath "/(qualifiertype)"
                                   qualifierName

WBEM-URI-UntypedNamespacePath    = namespacePath
WBEM-URI-UntypedClassPath        = namespacePath ":" className
WBEM-URI-UntypedInstancePath     = namespacePath ":"
                                   className "." key_value_pairs

namespacePath         = [namespaceType ":"] namespaceHandle
namespaceType         = ("http" ["s"]) / ("cimxml.wbem" ["s"])
```

```
// Note: The production rules for authority are defined in IETF
// RFC 3986 (Uniform Resource Identifiers (URI): Generic Syntax).

namespaceHandle       = ["//" authority] "/" [namespaceName]

// Note: IDENTIFIER is an identifier for CIM naming as defined in Appendix F
// (Unicode Usage) of DMTF DSP0004, CIM Infrastructure Specification.

namespaceName         = IDENTIFIER *("/"IDENTIFIER)

// Note: The production rules for stringValue, charValue, booleanValue,
// integerValue, and realValue are defined in Appendix A (MOF Syntax
// Grammar Description) of DMTF DSP0004, CIM Infrastructure Specification.
// The production datetimeValue is a datetime value as defined in
// Section 2.2.1 (Datetime Type) of DMTF DSP0004, CIM Infrastructure
// Specification.

// Untyped key value pairs

key_value_pairs       = key_value_pair *("," key_value_pair)
key_value_pair        = key_name "=" key_value
key_value             = stringValue / charValue / booleanValue /
                        integerValue / realValue /
                        "\"" datetimeValue "\"" /
                        "\"" referenceValue "\""

// Typed key value pairs

typed_key_value_pairs = typed_key_value_pair *("," typed_key_value_pair)
typed_key_value_pair  = key_name "=" typed_key_value
typed_key_value       = typed_string_value / typed_char_value /
                        typed_integer_value / typed_boolean_value /
                        typed_datetime_value / typed_real_value /
                        typed_reference_value
typed_string_value    = "(string)" stringValue
typed_char_value      = "(char16)" charValue
typed_boolean_value   = "(boolean)" booleanValue
typed_integer_value   = ("(uint8)" / "(sint8)" /"(uint16)"/ "(sint16)"/
                        "(uint32)" / "(sint32)" / "(uint64)" /
                        "(sint64)") integerValue
typed_real_value      = ("(real32)" / "(real32)") realValue
typed_datetime_value  = "(datetime)" "\"" datetimeValue "\""

// Note: A typed-reference-value shall consist of a reference-type designation,
// followed by a referenceValue enclosed in double quotes ("). A
// referenceValue shall be constructed by recursively following these rules to
// construct a reference to either a CIM Class or CIM Instance. Any double
// quote character or backslash character that appears in the referenceValue
// shall be escaped using a preceding backslash (\) character.

typed_reference_value = "(reference)" "\"" referenceValue "\""
```

5.7 WBEM URI Examples

5.7.1 Untyped Model Path

//www.acme.com/root/cimv2

//www.acme.com/root/cimv2:CIM_RegisteredProfile

https://jdd:test@acme.com:5959/cimv2:CIM_RegisteredProfile

https://jdd:test@acme.com:5959/cimv2:CIM_RegisteredProfile.InstanceID="acme:1"

5.7.2 Typed Model Path

//www.acme.com/root/cimv2/(namespace)

https://jdd:test@acme.com:5959/cimv2/(class)CIM_RegisteredProfile

https://jdd:test@acme.com:5959/cimv2/(instance)CIM_RegisteredProfile.InstanceID=(string)"acme:1"

https://jdd:test@acme.com:5959/cimv2/(qualifierType)Abstract

ANNEX A
(informative)

Change Log

Version	Date	Author	Description
1.0.0	July 29, 2009		DMTF Standard Release